# $Id: PKGBUILD 86001 2013-03-10 22:54:56Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Angel 'angvp' Velasquez <angvp[at]archlinux.com.ve>  
# Contributor: djgera
# Contributor: Patrick Melo <patrick@patrickmelo.eti.br>

pkgname=geany-plugins
pkgver=1.23
pkgrel=1
pkgdesc='Plugins for Geany'
arch=('x86_64' 'i686')
url='http://plugins.geany.org/'
license=('GPL')
depends=("geany=$pkgver" 'gtkspell' 'libwebkit' 'ctpl' 'gpgme' 'vte'
         'desktop-file-utils' 'gtk-update-icon-cache' 'lua51')
makedepends=('intltool' 'vala' 'gdb' 'cppcheck')
optdepends=('hspell: hebrew spell checker')
options=('!libtool')
install="$pkgname.install"
source=("http://plugins.geany.org/$pkgname/$pkgname-$pkgver.tar.bz2")
sha256sums=('ff26860edd1cbaea88c73bd8c2bd02f9ba04046674f05d3696baf66582e2bf58')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  # The geanylua plugin currently doesn't work with lua 5.2, only 5.1
  ./configure --prefix=/usr --libexecdir=/usr/lib
  make 
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
